
CONSTRUCTION OF A CGE MODEL FOR THE BUREAU OF
TRANSPORT ECONOMICS (R98/0013)

Progress Report 2 – 14 January 2001

Progress on Tasks: November 2000 to 14 January 2001

(1) Incorporate a multi-product refining industry

•  Basic MMRF-Green code modified to allow for multi-product industries – see
Appendix A;

•  Completed disaggregation of the existing commodity, 13 Petroleum products
(equivalent to ANZSIC classes 2510 and 2520) into six new commodities, PetrolAuto,
AvGasoline, AvTurbine, Diesel, LPG, PetrolOther (kerosene, heating oil and fuel oil) –
see Appendix B;

•  Basic MMRF-Green code modified to allow for specific taxes (federal excise, state
franchise licences) on each petroleum type – see Appendix C.

(2) Disaggregation of other transport

•  Existing industry and commodity 32 Other transport services (equivalent to ANZSIC
classes 62, 63, 64, 65, 66 and 67) split into four separate single-product, single-industry
components, RailTrans (ANZSIC 62), WaterTrans (63), AirTrans (64), OtherTrans
(65, 66 and 67).

(3) Improved treatment of motor vehicles in the household demand system

•  Implementation completed of Heuy-Lin's methodology. A new industry has been
created – PrivTranServ. This provides private transport services to households. It's
capital is the stock of motor vehicles used by households for transport. Its intermediate
inputs are fuel and materials (repairs, tyres, etc.) used to maintain and to run the private
vehicle fleet. The industry sells only to consumers. Consumers now do not directly buy
motor vehicles, nor do they buy the fuels and materials associated with vehicles.
Instead, they purchase private motor vehicle services, which effectively are
combinations of: the services provided by the vehicle stock; and of the fuel and
materials associated with vehicles.

(4) Implementation of GST

•  No progress.

(5) Enhanced treatment of transport

•  Disaggregation of non-road transport into sea, air and rail completed (see item (2)).

•  No work yet on distinguishing passenger/freight for all modes;

•  No work on incorporating different modal substitution possibilities for intra- and
interstate-freight.

Appendix A: Modifications to Allow for Multi-product Industries and Multi-
industry Products

The basic MMRF-Green code makes no allowance for multi-industry products or for
multi-product industries. Thus modifications are necessary to handle a multi-product
petroleum industry. These modifications allows the existing petroleum industry (13 Petroleum
products) to produce six products -

1. Petroleum for automotive use only (PetrolAuto),

2. Petroleum for aviation use only, commonly called AvGas (AvGasoline),

3. Aviation turbine fuel (AvTurbine),

4. Diesel for automotive use (Diesel),

5. LPG for automotive use (LPG),

6.  Other petroleum products, including kerosene, heating oil, fuel oil, paraffin wax, grease
base stock, petroleum jelly and petroleum solvents (PetrolOther).

The modifications are marked in the Tablo code by the keyword "mkmh". Below we
list all of the new and revised code introduced to handle the multi-product petroleum industry.
The code is listed in the order given in the Tablo file. Also included, where necessary, are
explanatory remarks.

A listing of new and revised code with explanatory remarks.

Excerpt A

Set
UPCOM # Single industry products #
(Agriculture, Forestry, IronOre, NonIronOre,
BlackCoal, Oil, NatGas, BrownCoal, Food, TCF, Woodpaper, Chemicals,
Nmet_prods, Cement, Steel, AlumMagnes, OthMet_prods, CarsParts,
Other_man, ElectBlack, ElectBrown, ElectGas, ElectOil, ElectOther,
ElectSupply, UrbanGasDis, Water, Construction, TradeHotels,
RoadTrans,RailTrans, WaterTrans, AirTrans, OtherTrans, Communic,
FinBusServ, Dwelling,PubServ, OthServ);

UPIND # Single product industries #
(Agriculture, Forestry, IronOre, NonIronOre, BlackCoal, Oil, NatGas,
BrownCoal, Food, TCF, Woodpaper, Chemicals, Nmet_prods, Cement, Steel,
AlumMagnes, OthMet_prods, CarsParts, Other_man, ElectBlack, ElectBrown,
ElectGas, ElectOil, ElectOther, ElectSupply, UrbanGasDis, Water,
Construction, TradeHotels, RoadTrans, RailTrans, WaterTrans, AirTrans,
OtherTrans, Communic, FinBusServ, Dwelling, PubServ, OthServ);

SUBSET UPCOM is subset of COM;
SUBSET UPIND is subset of IND;
UPCOM is subset of UPIND;
UPIND is subset of UPCOM;

Set
JPCOM # joint product commodities #  = COM - UPCOM;
JPIND # joint product industries #   = IND - UPIND;
!

This code introduces four new sets labelled UPCOM, UPIND, JPCOM and JPIND.
The first identifies the unique-industry commodities, while the second identifies the unique
industry products. The third set shows the commodities that are jointly produced by
industries. It is defined as the difference between the set COM (i.e., the set of all
commodities) and the set UPCOM. In the new implementation of the model, JPCOM includes
the six petroleum refinery products, with element labels: PetrolAuto, AvGasoline, AvTurbine,
Diesel, LPG and PetrolOther. The final new set JPIND shows the industries that jointly
produce commodities. It is defined as the difference between the set IND and the set UPIND.
In the new implementation, JPIND has one element, labeled Petrol.

Excerpt B

! New variables for multi-product industries and multi-industry products
!
    (all,i,COM)(all,s,REGSOURCE)
x0com(i,s)      # output domestic good i, region s #;
    (all,j,IND)(all,q,REGDEST)
p1tot(j,q)      # cost of production ind j, region q #;
    (all,i,JPCOM)(all,j,JPIND)(all,q,REGDEST)
q1(i,j,q)       # Supplies of commodities by industries #;

Three new variables are required to handle the new multi-product industries and multi-
industry products:

1. x0com(i,s) is the percentage change in output of commodity i produced in region s;

2. p1tot(j,q) is the percentage change in basic price of output for industry j in region q; and

3. q1(i,j,q) is the percentage change in output of commodity i produced by industry j in
region q.

Note that the third variable is defined only for the multi-product industries and the multi-
industry products.

Excerpt C

! Subsection 2.7.12: Coefficient and formulae for multi-product
industries and multi-industy products !

Coefficient (all,i,COM)(all,j,IND)(all,q,REGDEST) MAKE(i,j,q);
Read MAKE from file MDATA header "MAKE";
Update
(all,i,JPCOM)(all,j,JPIND)(all,q,REGDEST) MAKE(i,j,q)=q1(i,j,q)*p0a(i,q);
(all,i,UPCOM)(all,j,UPIND)(all,q,REGDEST) MAKE(i,j,q)=z(j,q)*p0a(i,q);

Coefficient
     (all,i,COM)(all,q,REGDEST)
MAKE_COM(i,q) # Sales of commodity i (from make) #;
     (all,j,IND)(all,q,REGDEST)
MAKE_IND(j,q) # Costs of industry j (from make) #;
Formula
     (all,i,COM)(all,q,REGDEST)
MAKE_COM(i,q) = sum{j,IND,MAKE(i,j,q)};
     (all,j,IND)(all,q,REGDEST)
MAKE_IND(j,q) = sum{i,COM,MAKE(i,j,q)};

Coefficient
     (all,i,COM)(all,j,IND)(all,q,REGDEST)
MAKESHR_COM(i,j,q) # Share of commodity i in output of j #

(all,i,COM)(all,j,IND)(all,q,REGDEST)
MAKESHR_IND(i,j,q) # Share of industry j in sales of i #;
  ZERODIVIDE DEFAULT 0.0;
Formula
     (all,i,COM)(all,j,IND)(all,q,REGDEST)
MAKESHR_COM(i,j,q) = MAKE(i,j,q)/MAKE_IND(j,q);
  ZERODIVIDE DEFAULT 0.0;
Formula
     (all,i,COM)(all,j,IND)(all,q,REGDEST)
MAKESHR_IND(i,j,q) = MAKE(i,j,q)/MAKE_COM(i,q);
  ZERODIVIDE OFF;

Coefficient
(all,i,IND) SIGMA1OUT(i) # CET transformation elasticities#;
READ SIGMA1OUT From File MDATA Header "SCET";

This code contains new coefficients and associated read and formula statements.
Values for two of the new coefficients come from primary data stored on the header array file
with generic name MDATA. Values for the remaining coefficients are inferred from the
primary data. The primary coefficients are:

1. MAKE(i,j,q) is the value of commodity i produced by industry j in region q; and

2. SIGMA1OUT(j) is the elasticity of transformation assumed for the products produced by
industry j.

At present, we assume that the elatisticiy of transformation is 1.0 for all industries. This
means that if the producer price of, say, diesel increased by five per cent relative to the
average of producer prices for all petroleum products, then five per cent more diesel will be
produced relative to the overall production of petroleum products.

Values for the MAKE coefficients are updated through a simulation according to the
update statements following the declaration statement for the coefficient. The update formula
has two parts. The first is applied to elements in the set JPCOM and JPIND. The second is
applied to all other elements of COM and IND.

Formulae determine values for the remaining coefficients declared in this extract.
These coefficients show values for column and row sums of the MAKE matrix and useful
MAKE-matrix shares – the share of commodity i produced by industry j in region q, and the
share of industry j in the output of commodity in region q.

Excerpt D

! Following formula used to check for balanced database !
! SHOULD EVALUATE TO ZERO !
    (all,i,COM)(all,s,REGSOURCE)
LOSTGOODS(i,s) = SALES(i,s) - MAKE_COM(i,s) !mkmh was COSTS(i,s)!;
    (all,i,COM)
LOSTGOODS1(i) = sum(s,REGSOURCE, LOSTGOODS(i,s));
    (all,s,REGSOURCE)
LOSTGOODS2(s) = sum(i,COM, LOSTGOODS(i,s));

! SHOULD EVALUATE TO ZERO !
Coefficient
    (all,j,IND)(all,q,REGDEST)
PureProfits(j,q);
Formula
    (all,j,IND)(all,q,REGDEST)
PureProfits(j,q) = COSTS(j,q) - MAKE_IND(j,q);

Two coefficients are used to check that the database is balanced (i.e., that the value of
sales matches the value of costs for all industries and commodities). LOSTGOODS(i,s) is
defined as the difference between the total basic value of sales of commodity i in region s as
implied by data in the BAS1, BAS2, ... matrices and the sales value of (i,s) implied by the
MAKE data. PUREPROFITS(j,q) is defined as the difference between the total basic value
costs for industry j in regions q as implied in the core data and the value of costs for (j,q)
implied by the MAKE data. Both coefficients should evaluation to zero (or very close to zero)
for all elements if the database is balanced.

Excerpt E

E_x0comA !mkmh! # Demand equals supply for margin commodities #
          (all,r,MARGCOM)(all,s,REGSOURCE)
(TINY+SALES(r,s))*x0com(r,s) !mkmh! =
    sum(j,IND, sum(q,REGDEST,  BAS1(r,s,j,q)*x1a(r,s,j,q)

..........................

E_x0comB !mkmh! # Demand equals supply for non margin commodities #
          (all,r,NONMARGCOM)(all,s,REGSOURCE)
(TINY+SALES(r,s))*x0com(r,s) =
        sum(j,IND, sum(q,REGDEST,  BAS1(r,s,j,q)*x1a(r,s,j,

.......................

This extract shows only portions of the market clearing conditions for margin and non-
margin commodities. They have been modified so that the left-hand variables are the outputs
of commodities, not the outputs of industries.

Excerpt F

E_z    # Average price received by industries #
    (all,j,IND)(all,q,REGDEST)
p1tot(j,q) = Sum(i,COM, MAKESHR_COM(i,j,q)*p0a(i,q));

E_q1    # Supplies of commodities by industries #
    (all,i,JPCOM)(all,j,JPIND)(all,q,REGDEST)
q1(i,j,q) = z(j,q) + SIGMA1OUT(j)*[p0a(i,q) - p1tot(j,q)];

E_p0aA  # Total output of domestic commodities #
    (all,i,JPCOM)(all,q,REGDEST)
x0com(i,q) = Sum(j,JPIND, MAKESHR_IND(i,j,q)*q1(i,j,q));

E_p0aB  # Total output of domestic commodities #
    (all,i,UPCOM)(all,q,REGDEST)
x0com(i,q) = Sum(j,UPIND, MAKESHR_IND(i,j,q)*z(j,q));

! Subsection 2.8.13: Basic prices !

E_p1cap !mkmh! # Zero pure profits in current production #
          (all,j,IND)(all,q,REGDEST)
(TINY+COSTS(j,q))*{p1tot(j,q) !mkmh was p0a(j,q)! - a(j,q)} =
    sum(i,COM,sum(s,ALLSOURCE,PVAL

..........................

The first equation in this excerpt explains the percentage change in the basic price of
output of industry j in region q as a weighted average of the percentage changes in basic
prices of commodities produced by j in q.

The second equation explains the percentage change in output of i by industry j in
region q. It says that, in the absence of price changes, the output of i by j in q will expand with
the overall level of activity of j in q. However, if the price of i increases relative to the average
price of j's output, then j's output of i will increase more quickly than j's overall level of
activity (i.e., industry j will transform its product mix in favour of i).

The third and fourth equations explain the percentage changes in output of commodity
i in region q as weighted averages across producers of the percentage changes in production
of i in q.

The final equation (only partially shown) is the representation of the pure-profits
condition. It has been modified on the left-hand side by replacing the commodity-based price
with the average industry price.

Appendix B: Modifications to Allow for Multi-product Industries and Multi-
industry Products

In this appendix, we describe the disaggregation of MMRF-Green commodity 13
Petroleum products into six components in the 1993-94 MMRF-Green database. The six
components are:

1. Petroleum for automotive use only (PetrolAuto),

2. Petroleum for aviation use only, commonly called AvGas (AvGasoline),

3. Aviation turbine fuel (AvTurbine),

4. Diesel for automotive use (Diesel),

5. LPG for automotive use (LPG),

6. Other petroleum products, including kerosene, heating oil, fuel oil, paraffin wax, grease
base stock, petroleum jelly and petroleum solvents (PetrolOther).

Step 1: Initial Disaggregation of flows to current production and to final demand

Our primary source of data is unpublished ABS statistics showing detailed commodity
sales by Input/Output user at the seven digit level of the Input-Output Commodity
Classification (IOCC).1 These data are summarised in Australian National Accounts Input-
Output Tables (Commodity Details) 1993-94 (ABS Catalogue number 5215.0). The detailed
ABS data split sales of Petroleum and coal products into sales of
•  25100010 Automotive petrol; gasoline refining; motor spirit (incl aviation spirit);
•  25100020 Kerosene (incl kerosene type jet fuel);
•  25100030 Gas oil or fuel oil (excl motor spirit and kerosene);
•  25100040 Petroleum bitumen; residues of petroleum oils and bituminous minerals;
•  25100053 Liquefied petroleum gas produced at refineries;
•  25100056 Refinery products nec;
•  25200010 Petroleum and coal products nec;
•  25201920 Other income;
•  25201950 Increase in stocks - work-in-progress; and
•  25209998 Petroleum and coal products nec.

These sales are spread across the standard Input-Output classification of purchasers – 107
industries and 7 final users.

                                                       
1 Another primary source of data for sales by user is ABARE (1999), Energy Projections to 2014-15, Tables
C and D. These show the usage (PJ) by broadly classified industries of each fuel type. The industry
classification is based on ANZSIC with the addition of a "residential" category covering the usage of fuels by
households. However, close examination of ABARE's data indicate a number of inconsistencies with
Input/output practices. Most importantly, the ABARE numbers show that the only users of Gasoline are the
"transport industries". Agricultural industries, for example, are not shown as using petroleum. In the input-
output data, agricultural industries are shown as purchasing a relatively large amount of Petroleum products.
Some of these purchases will be of gasoline for on-farm use or for use by owner-operators undertaking off-
farm business.

If we were to use ABARE's method of allocating automotive-fuel use, then considerable re-arranging of the
initial aggregated MMRF-Green data would be necessary before the disaggregation of Petroleum products
could take place. This is undesirable. Accordingly the ABS data are treated as the primary source of
information, with the ABARE data used for checking purposes.

The raw ABS data were processed in two stages. In the first stage, the data were re-
assembled into sales to users identified in MMRF-Green. This required mapping each of the
107 ABS industries to one of the 37 industries recognised in our model. Most of the mappings
were either one-to-one or involved clean aggregations of the ABS-defined users. However,
some disaggregations of the ABS data were necessary. These applied to:

1. the ABS industry Oil, gas and coal, which had to be split into the MMRF-Green
industries: 5 Black coal, 6 Crude oil, 7 Natural gas and 8 Brown Coal; and

2. the ABS industry Electricity, which had to be split into 21 Electricity – black coal, 22
Electricity – brown coal, 23 Electricity –gas, 24 Electricity – oil prods, 25 electricity –
other and 26 Electricity supply.

In the second processing stage, each of the detailed ABS commodities were mapped
to one or more of the new disaggregated petroleum products in MMRF-Green. The following
mappings (ABS to MMRF-Green) were used:

•  25100010 ⇒ PetrolAuto (100% all users except air transport, 20% air transport);
•  25100010 ⇒ AvTurbine (80% air transport);
•  25100020 ⇒ PetrolOther (100% all users except air transport, 0% air transport);
•  25100020 ⇒ AvGasoline (10% air transport);
•  25100020 ⇒ AvTurbine (90% air transport);
•  25100030 ⇒ Diesel (100% all users);
•  25100040 ⇒ PetrolOther (100% all users);
•  25100053 ⇒ LPG (100% all users);
•  25100056 ⇒ PetrolOther (100% all users);
•  25200010 ⇒ PetrolOther (100% all users);
•  25201920 ⇒ PetrolOther (100% all users);
•  25201950 ⇒ PetrolOther (100% all users);
•  25209998 ⇒ PetrolOther (100% all users);

Similar procedures were used to re-assemble the detailed ABS data on flows to final
demand into data on final demand for the six MMRF products.

Step 2: Inspection and adjustments of flows to current production and final demand

The disaggregation to this point suggests total usage of each of the (non-other)
petroleum products to be:

PetrolAuto $4581.4 million 48.1%

AvGasoline $71.2 million 0.7%

Avturbine $1334.4 million 14.0%

Diesel $3459.0 million 36.3%

LPG $77.0 million 0.8%

A check on these shares are provided by ABARE data on Energy consumption in Australia,
by industry and fuel type (Table C1). These data imply total-usage (Pj) shares in 1993-94 of:

PetrolAuto 595Pj 46.6%

AvGasoline 16Pj 1.3%

Avturbine 144.0Pj 11.3%

Diesel 414Pj 32.4%

LPG 107.7Pj 8.4%.



9

The shares in both datasets line up fairly closely for all commodities other than LPG. The
ABS data appear to grossly underestimate LPG usage. To correct for this, we deduct $500
million from household consumption of PetrolAuto and add it to household consumption of
LPG. After this adjustment, the shares in the ABS-based data are 42.9, 0.7, 14.0, 36.3 and 6.1.
These are judged to be more in line with the ABARE numbers.

Step 3: Final disaggregation of flows to current production and final demand

After step 2 we are left with a matrix of 6 rows (corresponding to the disaggregated
petroleum products) and 41 columns (corresponding to the 37 industry and 4 final users in our
model). From this matrix we construct column shares that sum to 1. These shares are then
used to disaggregate the existing MMRF-Green data for sales of petroleum products. We
apply these shares uniformly across each region.

Step 4: Disaggregation of margins

The same pro-rating procedure is used to separate the margins on flows of petroleum
products in the existing database into margins on the flows of each of the disaggregated
commodities. For example, suppose of total sales of petroleum products to agriculture 60 per
cent is of Diesel. Then 60 per cent of any margin on the flow of the aggregated commodity to
agriculture is allocated to the flow of diesel to agriculture.

 Step 5: Initial disaggregation of sales taxes

Initially, the sales-tax matrices were disaggregated in the same way as the basic and
margin flows. For example, 60 per cent of the taxes paid on the flow of petroleum products to
agriculture were allocated to the flow of diesel to agriculture.

 Step 6: Final adjustments to sales taxes

Inspection of the resulting tax data revealed a number of anomalies. Especially
worrying were:

1. differences in rates of ad valorem taxes paid by different users for the same petroleum
product in each state; and

2. disparities between the values for total taxes collected implied by the disaggregated data
and values reported in federal and state government budget papers.

Consequently, a number of ad hoc djustments were made to the tax matrices. The first
adjustments ensured that the rates of tax paid on each petroleum product were the same across
all users in each state. The second adjustment ensured that tax collections in total lined up
with official revenue statistics. These adjustments relied on data in Australian Institute of
Petroleum (1997), Oil and Statistical Review, Table 14. We assume that the only taxes
applying to petroleum products are federal excise duty and state franchise license fees. For
diesel, we deducted the diesel rebate from gross collections of taxes on diesel to obtain net
collections.2

After these adjustments, we checked that the specific tax rates implied by the tax
revenues and the quantities of fuels consumed by fuel type in each state (taken from Institute
of Petroleum, 1997) lined up with recorded rates of taxes (federal plus state) on each fuel
type. The recorded rates were taken from Federal Budget Strategy and Outlook 1997-98,

                                                       
2 In doing so, we assumed that: for farming, forestry and commercial fishing, the rebate is 100 per cent; for
mining, the rebate is 92 per cent; and for public services (covering hospitals, nursing homes and other
institutions), the rebate is 77 per cent.

Budget Paper No. 1, pp. 5-11 and from the budget papers of individual state treasuries. The
check was satisfactory.

The table belows summaries the net ad valorem and specific tax rates for each
petroleum product by state in our final database. The numbers are weighted averages across
all industrial and final users in each state. For products other than diesel, the rates of tax are
the same across all users in each state. For diesel, the rates differ across users in line with
differences in rebate rates. For example, the rate of tax on diesel used in agriculture is
calculated to be zero in each state.

Average Rates of Ad valorem (%) and Specific Tax (cents/litre) Rates
by Petroleum Product and State: MMRF-Green database for 1993-94

NSW VIC QLD SA WA TAS NT ACT AUS

Av valorem rates (%)

PetrolAuto 117.1 123.2 98.3 115.0 115.0 114.9 114.7 114.9 115.0

AvGasoline 35.8 35.8 35.8 35.8 35.8 35.8 35.8 0.0 35.8

Avturbine 3.5 3.5 3.5 3.5 3.5 3.5 3.5 3.5 3.5

Diesel 65.3 72.9 46.3 56.9 51.3 41.4 40.4 50.1 59.9

LPG 0.0 0.0 0.2 0.0 0.2 0.2 0.0 0.0 0.1

PetrolOther 4.8 4.8 4.8 4.8 4.8 4.8 4.8 4.8 4.8

Specific rates (cents/litre)

PetrolAuto 36.1 38.0 30.3 35.5 35.5 35.5 35.4 35.5 35.5

AvGasoline 20.0 20.0 20.0 20.0 20.0 20.0 20.0 0.0 20.0

Avturbine 2.0 2.0 2.0 2.0 2.0 2.0 2.0 2.0 2.0

Diesel 26.2 29.2 18.6 22.8 20.6 16.6 16.2 20.1 24.0

LPG 0.0 0.0 0.2 0.0 0.2 0.2 0.0 0.0 0.1

PetrolOther 2.1 2.1 2.1 2.1 2.1 2.1 2.1 2.1 2.1

Appendix C: Modifications to Allow for Specific taxes (federal excise, state
franchise licences) on each petroleum type

The modifications are marked in the Tablo code by the keyword "pettax". Below we
list all of the new and revised code introduced to handle specific taxes on petroleum products.
The code is listed in the order given in the Tablo file. Also included, where necessary, are
explanatory remarks.

A listing of new and revised code with explanatory remarks.

Excerpt A

! Here we impose in the equations for commodity taxes, a new term
that provides the flexibility necessary when imposing a specific
tax on the usage of petroleum products. The components of petroleum
products are declared in the set PetProd !

Set PETPROD # Petroleum products #
 (PetrolAuto, AvGasoline, AvTurbine, Diesel, LPG, PetrolOther);
Subset PETPROD is subset of COM;
Set NONPETPROD = COM - PETPROD;

Coefficient (all,i,COM) ISPETPROD(i);
! This coefficient has the value 1 if i is in set PetProd, 0 otherwise !
Formula
 (all,i,PetProd)    ISPETPROD(i) = 1.0;
 (all,i,NonPetProd) ISPETPROD(i) = 0.0;

Mapping COM2PET from COM to PetProd;
Formula
 (all,i,PetProd)    COM2PET(i) =  $POS(i);
 (all,i,NonPetProd) COM2PET(i) = "PetrolAuto";!Arbitrary!

Variable ! added for Pettax !
(Change) (all,i,PetProd)(all,s,ALLSOURCE)(all,j,IND)(all,q,REGDEST)
     pettax1(i,s,j,q)# Change in AV(%) equivalent of specific tax, 1 #;
(Change) (all,i,PetProd)(all,s,ALLSOURCE)(all,q,REGDEST)
     pettax3(i,s,q)  # Change in AV(%) equivalent of specific tax, 3 #;
(Change) (all,i,PetProd)(all,s,REGSOURCE)
     pettax4(i,s)    # Change in AV(%) equivalent of specific tax, 4 #;

! At present, we assume that taxes on petroleum products are either
federal excise or state franchise licence fees. We make no allowance for
a GST !
(Change) (all,p,PETPROD)(all,q,REGDEST)
     del_revexise(p,q)  # Change in revenue from fed petroleum exise #;
(Change) (all,p,PETPROD)(all,q,REGDEST)
     del_revfran(p,q) # Change in revenue from state franchise-fees #;

In this excerpt, we define a number of sets, coefficients, mappings and variables that
will be used later in the code.

The set PETPROD is the set of all petroleum products. The set NONPETPROD is the
set of all non-petroleum products. The coefficient ISPETPROD(i), for all i in set COM, takes
the value of 1 for i in the set PETPROD, and zero otherwise. COM2PET(i), for all i in set

COM, is a mapping. It takes an element of the set COM as input and produces as output an
element of the set PETPROD.

In the remainder of the excerpt, we declare a number of tax-rate and tax-collection
variables relating to sales of petroleum products. The variables pettax1, pettax3 and pettax4
represent ordinary changes in ad valorem tax rates (%) on petroleum products used by
industries (pettax1), and consumers (pettax3) and sold for export (pettax4). We assume that
there are no taxes/sales of petroleum products to industries for investment, or to governments
for consumption. The variables del_revexcise and del_revran represent ordinary changes in
tax revenues ($million) from the collection of excise and state franchise fees on petroleum
products. In the database for 1993-94, it is assumed that taxes on petroleum products are
either federal excise or state franchise licence fees.

Excerpt B

Equation
E_deltax1 # Tax rate on sales to User 1 #
    (all,i,COM)(all,s,ALLSOURCE)(all,j,IND)(all,q,REGDEST)
deltax1(i,s,j,q) =
!pettax!(1-ISPETPROD(i))* {
deltax(i) + deltax1all + deltaxsource(s) + deltaxdest(q) +
deltax1_is(i,s)} +
!greenmod!ISFUEL(i)*fueltax1(COM2FUEL(i),s,j,q)+delcomtax(i,s)+deltax1_ij
(i,j) +
!pettax!ISPETPROD(i)*pettax1(COM2PET(i),s,j,q);

E_deltax3 # Tax rate on sales to User 3 #
    (all,i,COM)(all,s,ALLSOURCE)(all,q,REGDEST)
deltax3(i,s,q) =
!pettax!(1-ISPETPROD(i))* {
deltax(i) + deltax3all + deltaxsource(s)+deltaxdest(q)} +
!greenmod!ISFUEL(i)*fueltax3(COM2FUEL(i),s,q)+delcomtax(i,s)+deltax3comp
+
 deltax3_i(i) +
!pettax!ISPETPROD(i)*pettax3(COM2PET(i),s,q);

E_deltax4 # Tax rate on sales to User 4 #
    (all,i,COM)(all,s,REGSOURCE)
deltax4(i,s) =
!pettax!(1-ISPETPROD(i))* {
deltax(i)+deltax4all+deltaxsource(s)+deltaxdest("foreign")} +
!greenmod!delcomtax(i,s) +
!pettax!ISPETPROD(i)*pettax4(COM2PET(i),s);

In this excerpt we show existing equations amended by the inclusion of the petroleum
tax variables declared in excerpt A. These equations allow for flexibility in the handling of the
variables deltax1, deltax3 and deltax4. These are the core variables representing changes in ad
valorem rates of tax on sales. All three equations have been modified so that changes in ad
valorem rates of tax on petroleum products can only be affected through non-zero values for
the pettax variables.

Excerpt C

! SOFTY122 contains commodity tax revenue (other than import tariff
revenue) that accure to state and federal governments. !

Coefficient
    (all,q,DOMDEST)
C_SOFTY122a(Q) # Other commodity taxes - petroleum franchise fees #;
    (all,q,DOMDEST)
C_SOFTY122b(Q) # Other commodity taxes - petroleum excise duty #;
    (all,q,DOMDEST)
C_SOFTY122c(Q) # Other commodity taxes-non-petrol other commodity tax #;

Read
    (all,q,DOMDEST)
C_SOFTY122a(Q) from file NDATA header "G08a";
    (all,q,DOMDEST)
C_SOFTY122b(Q) from file NDATA header "G08b";
    (all,q,DOMDEST)
C_SOFTY122c(Q) from file NDATA header "G08c";

Variable
    (all,q,DOMDEST)
softy122c(q)  # Non-petroleum other commodity taxes #;

Update
(change) (all,q,REGDEST)
C_SOFTY122a(q) = Sum(p,PETPROD, del_revfran(p,q));
(change)
C_SOFTY122a("Federal") = 0.0;
(change) (all,q,REGDEST)
C_SOFTY122b(q) = 0.0;
(change)
C_SOFTY122b("federal") = Sum(p,PETPROD, Sum(q,REGDEST,
del_revexise(p,q)));
    (all,q,DOMDEST)
C_SOFTY122c(q) = softy122c(q);

In the existing section of code dealing with government financial transactions (the
"SOFTY" accounts), taxes collected on sales of petroleum products by individual
governments are combined with collections of other commodity taxes (except tariffs). The
value of these collections is given by the value of coefficient C_SOFTY122. The
corresponding variable is softy122. We now divide SOFTY122 into three parts:

1. SOFTY122a - Petroleum franchise fees accruing to state governments;

2. SOFTY122b - Petroleum exise duty accuring to the federal government; and

3. SOFTY122c - Non-petroleum commodity taxes (other than tariffs).

The percentage change in C_SOFTY122c is given by the variable softy122c. There are no
percentage change variables softy122a and softy122b. The change in C_SOFTY122a is given
by the value of del_revfran(p,q) summed over all petroleum products. The change in
C_SOFTY122b for the federal government is given by the value of del_revexise(p,q) summed
over all petroleum products and regions. Note that each of the coefficients is defined over the
set of DOMDEST, which covers all state/territory governments and the federal government.
However, there are no collections of franchise fees by the federal government and, similarly,
there no collections of excise taxes by the state/territory governments.

Excerpt D

Variable
fsofty122  #Endogenous shift variable, ensuring tax collection=tax rev #;

Equation
E_softy122cI # Non-petroleum other commodity taxes - regions #
    (all,q,REGDEST)
softy122c(q) = ti_r(q) + fsofty122;

E_softy122cII # Non-petroleum other commodity taxes - federal #
softy122c("federal") = ti + fsofty122;

E_softy122I # Other commodity taxes - regions #
     (all,q,REGDEST)
C_SOFTY122(q)*softy122(q) =
             100*Sum(p,PETPROD, del_revfran(p,q)) +
             C_SOFTY122c(q) * softy122c(q);

E_softy122II # Other commodity taxes - federal #
C_SOFTY122("federal")*softy122("federal") =
             100*Sum(p,PETPROD, Sum(q,REGDEST, del_revexise(p,q))) +
             C_SOFTY122c("federal") * softy122c("federal");

E_fsofty122
Sum(q,REGDEST, C_DOMPY320(q)*dompy320(q)) +
Sum(d,DOMDEST, C_SOFTQ300(d)*softq300(d)) =
                              Sum(s,DOMDEST, C_SOFTY122(s)*softy122(s));

These equations translate other-commodity tax revenue collected in each state into tax
revenue received by each state/territory government and by the federal government. The
amount of other-commodity tax collected in state q equals
(C_DOMPY320(q)+SOFTQ300(q)), where C_DOMPY320(q) is measured net of subsidies
and SOFTQ300(q) is the value of subsidies applying in region q. The percentage change in
dompy320 is ti_r. The Australia-wide percentage change in "other" commodity taxes is ti. To
ensure that:

Sum(q, C_DOMPY320(q)*dompy320(q)+C_SOFTQ300*softq300)
                              = Sum(s, DOMDEST, C_SOFTY122(s)*softy122(s)),

we enter this as an explicit equation (with name E_fsofty122) and introduce a shift variable
fsofty122 in the equation for softy122c(q) (E_softy122cI combined with E_softy122cII)
which is endogenised by the constraint.

Equation E_softy122I ensures that total other commodity taxes received by
state/territory government q includes all of the franchise licence fees collected on sales of
petroleum products in region q. Equation E_softy122II ensures that total other commodity
taxes received by the federal government includes all of the ecise duty collected on sales of
petroleum products in each region, but none of the frachise licence fees.

Excerpt E

Coefficient
    (All,p,PETPROD)(All,s,ALLSOURCE)(All,j,IND)(All,q,REGDEST)
QPET1(p,s,j,q) # Quantity (litre) of pet product sold to industry #;
    (All,p,PETPROD)(All,s,ALLSOURCE)(All,q,REGDEST)
QPET3(p,s,q)   # Quantity (litre) of pet product sold to consumption #;
    (All,p,PETPROD)(All,s,REGSOURCE)
QPET4(p,s)     # Quantity (litre) of pet product sold to export #;

Read
QPET1 from file GASDATA header "QPT1";
QPET3 from file GASDATA header "QPT3";
QPET4 from file GASDATA header "QPT4";

Update
    (All,p,PETPROD)(All,s,ALLSOURCE)(All,j,IND)(All,q,REGDEST)
QPET1(p,s,j,q) = x1a(p,s,j,q);
    (All,p,PETPROD)(All,s,ALLSOURCE)(All,q,REGDEST)
QPET3(p,s,q)   = x3a(p,s,q);
    (All,p,PETPROD)(All,s,REGSOURCE)
QPET4(p,s)     = x4r(p,s);

In this excerpt and in the two excerpts following, we put in place the code necessary to
handle specific ($ per litre) taxes on petroleum products. First we read in the quantities (litre)
of petroleum products sold to industries, to consumption and to export. These quantities are
updated using the core variables x1a, x3a and x4r.

Excerpt F

Variable
(change) (all,p,PETPROD)(all,s,ALLSOURCE)(all,j,IND)(all,q,REGDEST)
del_pettax1(p,s,j,q) # Change ($/litre) in specific tax on pet, user 1 #;
(change) (all,p,PETPROD)(all,s,ALLSOURCE)(all,q,REGDEST)
del_pettax3(p,s,q) # Change ($/litre) in specific tax on pet, user 3 #;
(change) (all,p,PETPROD)(all,s,REGSOURCE)
del_pettax4(p,s) # Change ($/litre) in specific tax on pet, user 4 #;
(change) (all,p,PETPROD)
del_petexise(p)  # Change ($/litre) in federal exise on pet product p #;
(change) (all,p,PETPROD)(all,q,REGDEST)
del_petfran(p,q) # Change ($/litre) in franchise fee on pet product p #;

These varibles represent ordinary changes in specific taxes ($/litre) on petroleum
products. The last two variables are the change in the rate of federal exise on petrolum
product p and the change in the rate of franchise fee paid on petroleum product p in
state/territory q.

Excerpt G

Equation
E_del_pettax1 # General treatment for del_pettax1 #
    (all,p,PETPROD)(all,s,ALLSOURCE)(all,j,IND)(all,q,REGDEST)
del_pettax1(p,s,j,q) = del_petexise(p) + del_petfran(p,q) ;

E_del_pettax3 # General treatment for del_pettax3 #
    (all,p,PETPROD)(all,s,ALLSOURCE)(all,q,REGDEST)
del_pettax3(p,s,q) = del_petexise(p) + del_petfran(p,q) ;

E_del_pettax4 # General treatment for del_pettax4 #
    (all,p,PETPROD)(all,s,REGSOURCE)
del_pettax4(p,s) = del_petexise(p) + del_petfran(p,s) ;

E_pettax1 # Change in AV(%) equivalent of specific petrol tax, user 1 #
    (all,p,PETPROD)(all,s,ALLSOURCE)(all,j,IND)(all,q,REGDEST)
(tiny+BAS1(p,s,j,q))*pettax1(p,s,j,q) =
           ((100/1000000)*QPET1(p,s,j,q)*ENERINDEX)*del_pettax1(p,s,j,q)+
                                TAX1(p,s,j,q)*(gastaxindex-p0a(p,s));

E_pettax3 # Change in AV(%) equivalent of specific petrol tax, user 3 #
    (all,p,PETPROD)(all,s,ALLSOURCE)(all,q,REGDEST)
(tiny+BAS3(p,s,q))*pettax3(p,s,q) =
           ((100/1000000)*QPET3(p,s,q)*ENERINDEX)*del_pettax3(p,s,q) +
                                TAX3(p,s,q)*(gastaxindex-p0a(p,s));

E_pettax4 # Change in AV(%) equivalent of specific petrol tax, user 4 #
    (all,p,PETPROD)(all,s,REGSOURCE)
(tiny+BAS4(p,s))*pettax4(p,s) =
           ((100/1000000)*QPET4(p,s)*ENERINDEX)*del_pettax4(p,s) +
                                      TAX4(p,s)*(gastaxindex-p0a(p,s));

There are two sets of equations in this excerpt. The first set allow for a general
treament of the del_pettax variables. They respond only to changes in the federal excise rate
or to changes in state franchise fee rates.

The second set of equations express changes in the specific taxes on petroleum
products as changes in the corresponding ad valorem rates of tax. The basic equation is:

1000000/100/ SIQVQP ××=××

where: P is the basic price per litre ($million/litre);

Q is the quantity (litre) of petroleum product sold

V is the percentage ad valorem tax rate on the petroleum product ( a number like 10, 
not 0.1);

S is the specific tax ($/litre); and

I is the general price index to which the specific tax is indexed.

In change form, we write

piSdelSVdelV −+×=× _)/100(_)/100( .

V is computed as 100*TAX/BAS, where TAX and BAS refer to the tax and basic flow
components of the core GE data. S is computed as 1000000*TAX/(Q*I). We use the same
index price as used for the specific carbon tax (ie the national CPI). Q is read from the
GASDATA database (see the previous excerpt).

Excerpt H

Variable
(change) (all,p,PETPROD)(all,q,REGDEST)
del_QPETSU(p,q) # Change in quantity (litres) of pet product p in q #;
(change) (all,p,PETPROD)
del_QPETSUD(p)  # Change in quantity (litres) of pet product p #;
(change)
del_QPETTOT     # Total change in quantity (litres) of pet products #;
    (all,p,PETPROD)(all,q,REGDEST)
v_QPETSU(p,q)   # % Change in quantity of pet product p in q #;
    (all,p,PETPROD)
v_QPETSUD(p)    # % Change in quantity of pet product p #;
v_QPETTOT       # % Total change in quantity of pet products #;

Equation
E_del_qpetsu
    (all,p,PetProd)(all,q,REGDEST)
del_QPETSU(p,q) = 0.01*{
             Sum(s,ALLSOURCE, Sum(j,IND, QPET1(p,s,j,q)*x1a(p,s,j,q))) +
               Sum(s,ALLSOURCE,            QPET3(p,s,q)*x3a(p,s,q)) +
                                           QPET4(p,q)*x4r(p,q) } ;

E_del_qpetsud
    (all,p,PETPROD)
del_QPETSUD(p) = Sum(q,REGDEST, del_QPETSU(p,q)) ;

E_del_qpettot
del_QPETTOT = Sum(p,PETPROD, del_QPETSUD(p));

E_v_qpetsu
    (all,p,PetProd)(all,q,REGDEST)
(tiny+QPET_SU(p,q))*v_QPETSU(p,q) = {
              Sum(s,ALLSOURCE, Sum(j,IND, QPET1(p,s,j,q)*x1a(p,s,j,q))) +
               Sum(s,ALLSOURCE,            QPET3(p,s,q)*x3a(p,s,q)) +
                                           QPET4(p,q)*x4r(p,q) } ;

E_v_qpetsud
    (all,p,PETPROD)
(tiny+QPET_SUD(p))*v_QPETSUD(p) = Sum(q,REGDEST,
QPET_SU(p,q)*v_QPETSU(p,q));

E_v_qpettot
(tiny+QPET_TOT)*v_QPETTOT = Sum(p,PETPROD, QPET_SUD(p)*v_QPETSUD(p));

The equations in this excerpt define some summary variables for ordinary and
percentage changes in the volume of petroleum products.

Excerpt I

Coefficient
    (all,p,PETPROD)(all,q,REGDEST)
REVEXISE(p,q)  # Revenue ($m) from federal exise on pet product p #;
    (all,p,PETPROD)(all,q,REGDEST)
REVFRAN(p,q) # Revenue ($m) from state franchise-fee on pet prod p #;

Read REVFRAN from file GASDATA header "FRAN";
Update
(change)(All,p,PETPROD)(All,q,REGDEST)
REVFRAN(p,q) = del_revfran(p,q);
Formula
(all,p,PETPROD)(all,q,REGDEST)
REVEXISE(p,q)  = Sum(s,ALLSOURCE, Sum(j,IND, TAX1(p,s,j,q))) +
                 Sum(s,ALLSOURCE,            TAX3(p,s,q)) +
                                             TAX4(p,q) -
                                             REVFRAN(p,q);

Equation ! Revenue ($m) = QIS/1000000; 100delR/R = 100delS/s + i + q !
E_del_revfran
    (all,p,PETPROD)(all,q,REGDEST)
100*del_revfran(p,q) =
            {100*ENERINDEX*QPET_SU(p,q)/1000000}*del_petfran(p,q) +
                             REVFRAN(p,q)*(v_qpetsu(p,q) + gastaxindex);

E_del_revexise ! Revenue ($m) = QIS/1000000; 100delR/R = 100delS/s + i +
q !
    (all,p,PETPROD)(all,q,REGDEST)
100*del_revexise(p,q) =
            {100*ENERINDEX*QPET_SU(p,q)/1000000}*del_petexise(p) +
                             REVEXISE(p,q)*(v_qpetsu(p,q) + gastaxindex);

In this final piece of code, we explain the changes in revenue ($million) from federal
excise on petroleum products and from state frachise licence fees on petroleum products.


